

## GB002: 2次元静電場

本 whitepaper は Gunnar Backstrom 氏の承諾のもと、書籍“*Simple Fields of Physics by Finite Element Analysis*” に記されている多数の FlexPDE 適用事例の中からその一部を紹介するものです。

本 whitepaper では 2 次元静電場を表すラプラス方程式を種々の境界条件の下で解き、電場の様子を調べてみることにします。今、静電ポテンシャルを  $U$ 、電場ベクトルを  $E$  とすると、真空中の定常状態では

$$E = -\nabla U \quad (1)$$

$$\nabla \cdot E = 0 \quad (2)$$

という数式が成り立ちます。(2) は電荷が存在しない状態での電場を規定するマックスウェルの方程式です。この 2 式より  $U$  に関するラプラス方程式


$$\Delta U \equiv \nabla \cdot \nabla U \equiv \text{div}(\text{grad}(U)) = 0 \quad (3)$$

が導かれます。 $U(x, y)$  を求めることができれば (1) を使って  $E(x, y)$  を算出することができます。

## 1 Case1 - 金属箱中の金属棒

ここでは断面が正方形の金属箱の中心部に断面が円形の金属棒がつるされている状態を考えます。金属箱も金属棒も  $z$  軸方向には十分な長さを持つものとし、また金属棒はその両端を絶縁性のワイヤで固定されているものとし、右の図はその金属箱と金属棒の中央付近での断面図で、正方形の 1 辺の長さは 2、円の半径は 0.3 とします。

外側の金属箱の電位は 0、内側の金属棒の電位は 1 に保たれているとしたとき、右図のグレイで塗った領域内での静電場の様子を FlexPDE で調べてみましょう。



## 1.1 Problem descriptor [ estatic01a.pde ]

まずタイトルを設定します。

```
TITLE
'Metal Rod in a Box' { estatic01a.pde }
```

次に演算精度に関するセクタをセットします。デフォルトは 0.002 なのですが、ここでは精度を 0.001 とします。

```
SELECT
Errlim = 1e-3
```

従属変数を定義します。

```
VARIABLES
U { Electric potential }
```

偏微分方程式の定義に先立ち、パラメータ類を定義します。これらは境界の定義、及び境界条件の設定に際して使用されます。

```
DEFINITIONS
L = 1    r0 = 0.3    U_rod = 1.0
Ex = -dx(U)    Ey = -dy(U)
E = -grad(U)    Em = magnitude(E)
```

ラプラス方程式を定義します (座標系については定義していないので 2 次元直交座標系が仮定されます)。

```
EQUATIONS
div(grad(U)) = 0
```

次に境界の形状と境界条件を設定します。今回の境界条件は外側境界、内側境界、共に Dirichlet 型であるため、Value 文を使用します。なお、境界に対して 'box' と 'rod' という名称を設定していますが、それらは PLOTS セクションの中で使用されることになる点に注意してください。

```
BOUNDARIES
Region 1
Start 'box' (-L, -L)
Value(U) = 0 Line to (L, -L) to (L, L) to (-L, L) to Close
Start 'rod' (r0, 0)
Value(U) = U_rod Arc(Center = 0,0) Angle = 360
```

最後に出力すべき情報を指定します。

**PLOTS**


```
Grid(x, y)
Contour(U)   Surface(U)
Elevation(U) on 'box'   Elevation(U) on 'rod'
Contour(Em)   Elevation(Em) on 'box'
Vector(E) norm
```

**END**

## 1.2 実行結果


### (1) Grid(x, y)

FlexPDE によって設定されたグリッドの形状を示しています。メッシュ密度に目立った変化は見られません。




## (2) Contour(U)

解析対象領域（ドメイン）上での関数  $U(x, y)$  の等高線図、すなわち等電位線は次のようになります。‘rod’上で  $U = 1$ 、‘box’上で  $U = 0$  となっている点に注意してください。




## (3) Surface(U)

関数  $U(x, y)$  の曲面の形状をプロットしたものです。




(4) Elevation( $U$ ) on 'box'

$U(x, y)$  の値を境界 'box' 上でプロットしたものです。指定値通り、境界 'box' 上では  $U$  の値は 0 となっています。なお、境界線上の点との対応は赤の数字で示されています。


(5) Elevation( $U$ ) on 'rod'

$U(x, y)$  の値を境界 'rod' 上でプロットしたものです。指定値通り、境界 'rod' 上では  $U$  の値は 1 となっています。




## (6) Contour(Em)

電場ベクトル  $E$  の絶対値（大きさ）に関する等高線図を示したものです。




## (7) Elevation(Em) on 'box'

電場ベクトル  $E$  の大きさを境界 'box' 上でプロットしたものです。各辺の中央部で  $|E|$  の値は最も大きくなっています。



## (8) Vector(E) norm


これは電場ベクトル  $E$  のベクトル場をプロットしたものです。norm という修飾子を指定しているため、矢印の長さは等長となります。ベクトル場の向きはこれによってより明確なものとなりますが、大きさについては色で判断することになります。金属棒表面から放射状に放たれたベクトルは金属箱の近傍で向きを変え、その表面には垂直な形で入射している点に注意してください。



## 2 Case2 - 円筒中の金属棒

今度は外側が円筒形の金属製チューブで、その中に断面が正方形の金属棒が置かれている状況について調べてみましょう。円の半径は1、正方形の辺の長さは0.6であるとして、電位の設定はCase1の場合と同じで、外側のチューブは電位0に、中側の金属棒は電位1に保たれている状態を想定します。

スクリプトはCase1で作成した estatic01a.pde のわずかな変更で済みますが、ここでは念のため全体を再掲しておきます。



## 2.1 Problem descriptor [ estatic01b.pde ]

基本形は Case1 で作成した estatic01a.pde と変わりません。

```

TITLE
  'Metal Bar in a Tube'      { estatic01b.pde }

SELECT
  Errlim = 1e-3

VARIABLES
  U                          { Electric potential }

DEFINITIONS
  r0 = 1.0    b = 0.3    U_bar = 1.0
  Ex = -dx(U)  Ey = -dy(U)
  E = -grad(U)  Em = magnitude(E)

EQUATIONS
  div(grad(U)) = 0

BOUNDARIES
  Region 1
    Start 'tube' (r0, 0)
      Value(U) = 0  Arc(Center = 0,0) Angle = 360
    Start 'bar' (-b, -b)
      Value(U) = U_bar  Line to (b, -b) to (b, b) to (-b, b) to Close

PLOTS
  Grid(x, y)
  Contour(U)    Surface(U)
  Elevation(U) on 'tube'    Elevation(U) on 'bar'
  Contour(Em)    Surface(Em)
  Elevation(Em) on 'tube'    Elevation(Em) on 'bar'
  Vector(E) norm

END


```



## 2.2 実行結果


### (1) Grid(x, y)

FlexPDE によって自動生成されるメッシュは金属棒の 4 隅の部分を中心にかなり密度の高いものとなっています。



### (2) Contour(U)


解析対象領域（ドメイン）上での関数  $U(x, y)$  の等高線図、すなわち等電位線は次のようになります。‘bar’ 上で  $U = 1$ 、‘tube’ 上で  $U = 0$  となっている点に注意してください。



## (3) Surface(U)

関数  $U(x, y)$  の曲面の形状をプロットしたものです。

Metal Bar in a Tube


15:27:43 6/18/08  
FlexPDE 5.0.22

export01b: Grid#3 p2 Nodes=1495 Cells=707 RMS Err= 6.7e-4  
Integral= 0.958715

## (4) Elevation(U) on 'tube'

$U(x, y)$  の値を境界 'tube' 上でプロットしたものです。一見指定値 0 から大きくはずれているように見えるかも知れませんが、スケールが  $e-16$  である点にご注意ください。


Metal Bar in a Tube

15:27:43 6/18/08  
FlexPDE 5.0.22

export01b: Grid#3 p2 Nodes=1495 Cells=707 RMS Err= 6.7e-4  
Integral= -5.114888e-16


## (5) Elevation(U) on 'bar'

$U(x, y)$  の値を境界 'bar' 上でプロットしたものです。指定値通り、境界 'bar' 上では  $U$  の値は 1 となっています。




## (6) Contour (Em)

電場ベクトル  $E$  の絶対値 (大きさ) に関する等高線図を示したものです。金属棒の 4 隅に鋭いピークが存在する点に注意してください。




(7) Surface(Em)

電場ベクトル  $E$  の大きさに関する曲面図です。等高線図よりピークの存在が明確です。電場中に置かれた鋭い金属片の周囲では放電が発生しやすくなるため、現実問題としても重要な意味を持ってきます。




(8) Elevation(Em) on 'tube'

電場ベクトル  $E$  の大きさを外側の境界である 'tube' 上でプロットしたものです。境界上の点との位置関係は赤字で示されています。




## (9) Elevation(Em) on 'bar'

同様に電場ベクトル  $E$  の大きさを内側の境界である 'bar' 上でプロットしたものです。4 隅に鋭いピークが現れています。



## (10) Vector(E) norm

電場ベクトル  $E$  のベクトル場をプロットしたものです。norm という修飾子を指定しているため、矢印の長さは等長となります。ベクトル場の向きはこれによってより明確なものとなりますが、大きさについては色で判断することになります。



FlexPDE のステータスウィンドウに表示されている Max Error の値が指定された誤差許容値 (Errlim) の範囲を超えている点に注意してください。本来であればその範囲におさまるようメッシュの細分化が行われるのですが、ノード数にも上限があるため、このような値に留まる結果となっています。

### 3 Case3 - 平行場に置かれた金属棒

今度は  $E_0$  という平行な静電場の中に断面が円形の金属棒を配置したときに、その周辺で電場がどう変化するかを調べてみます。解析対象のドメインとしては Case1 の例と同じものを使用しますが、境界条件の設定が全く異なったものとなります。すなわちドメイン境界上で  $U$  の外向き法線微分の値を指定します。すなわち上辺と下辺上では

$$\frac{\partial U}{\partial n} = 0$$


ですが、右辺上では

$$\frac{\partial U}{\partial n} = \frac{\partial U}{\partial x} = -(\mathbf{E}_0)_x$$

を、左辺上では

$$\frac{\partial U}{\partial n} = \frac{\partial U}{\partial(-x)} = (\mathbf{E}_0)_x$$

を指定することになります。なお、金属棒表面上での電位は 0 とします。



#### 3.1 Problem descriptor [ estatic01c.pde ]

Case1 で作成した estatic01a.pde をベースに次のようなスクリプトを用意します。

TITLE

```
'Metal Rod across a Parallel Field' { estatic01c.pde }
```

SELECT

```
Errlim = 1e-4
```

VARIABLES

```
U { Electric potential }
```

## DEFINITIONS

```
L = 1    r0 = 0.3    E0x = 1e3
Ex = -dx(U)    Ey = -dy(U)
E = -grad(U)    Em = magnitude(E)
```

## EQUATIONS

```
div(grad(U)) = 0
```

## BOUNDARIES

```
Region 1
Start 'outer' (-L, -L)
Natural(U) = 0    Line to (L, -L)
Natural(U) = -E0x Line to (L, L)
Natural(U) = 0    Line to (-L, L)
Natural(U) = E0x  Line to Close
Start 'rod' (r0, 0)
Value(U) = 0      Arc(Center = 0,0) Angle = 360
```

## PLOTS


```
Grid(x, y)
Contour(U)    Surface(U)
Elevation(U) on 'outer'    Elevation(U) on 'rod'
Contour(Em)    Vector(E)
Elevation(Ex) on 'outer'    Elevation(Em) on 'rod'
```

```
END
```

## 3.2 実行結果


## (1) Grid(x, y)

FlexPDE によって自動生成されたメッシュの構成を次に示します。



## (2) Contour (U)

解析対象領域 (ドメイン) 上での関数  $U(x, y)$  の等高線図、すなわち等電位線は次のようになります。






(3) Surface(U)

関数  $U(x, y)$  の曲面の形状をプロットしたものです。

Metal Rod across a Parallel Field

15:58:30 6/18/08  
FlexPDE 5.0.22




export01c: Grid#2 p2 Nodes=1537 Cells=725 RMS Err= 5.6e-5  
Integral= 9.830189e-3

(4) Elevation(U) on 'outer'

$U(x, y)$  の値を外側の境界 'outer' 上でプロットしたものです。

Metal Rod across a Parallel Field


15:58:30 6/18/08  
FlexPDE 5.0.22



export01c: Grid#2 p2 Nodes=1537 Cells=725 RMS Err= 5.6e-5  
Integral= 5.672875e-3


## (5) Elevation(U) on 'rod'

$U(x, y)$  の値を内側の境界 'rod' 上でプロットしたものです。一見指定値 0 から大きくはずれているように見えるかも知れませんが、スケールが  $e-12$  である点にご注意ください。




## (6) Contour (Em)

電場ベクトル  $E$  の絶対値（大きさ）に関する等高線図を示したものです。金属棒の左右端で  $|E|$  の値は最も大きく、上下端ではほとんど 0 となっている点にご注意ください。




## (7) Vector(E)

電場ベクトル  $E$  のベクトル場をプロットしたものです。電場ベクトル  $E$  は静電ポテンシャル  $U$  の勾配 (gradient) として導かれるわけですが、その絶対値  $|E|$  が 'rod' の上下端で 0 となること、及び左右端で最大となることは (3) の曲面図の形状、勾配から読み取れます。



## (8) Elevation(Ex) on 'outer'

$E_x (= -\frac{\partial u}{\partial x})$  の値を外側の境界上でプロットしたものです。左右の境界上で指定値どおりであることを確認ください。なお (4) の elevation プロットに見られるように、 $U$  の値は左右の境界上で一定ではありません。



## (9) Elevation(Em) on 'rod'

電場ベクトル  $E$  の大きさを内側の境界である 'rod' 上でプロットしたものです。

